Title stata.com

```
graph twoway function — Twoway line plot of function
```

Description Quick start Menu Syntax
Options Remarks and examples Reference Also see

Description

```
twoway function plots \mathbf{y}=f(\mathbf{x}), where f(\mathbf{x}) is some function of \mathbf{x}. That is, you type . twoway function \mathbf{y}=\mathbf{sqrt}(\mathbf{x})
```

It makes no difference whether y and x are variables in your data.

Quick start

```
Graph the function \ln\{x/(1-x)\}\ from 0 to 1
     twoway function y = ln(x/(1-x))
Same as above
     twoway function y = logit(x)
Graph the function y = x^2 from -1 to 1
     twoway function y = x^2, range(-1 1)
Same as above, but as a horizontal graph
     twoway function y = x^2, range(-1 1) horizontal
Probability density function of Student's t distribution with 4 degrees of freedom
     twoway function y = tden(4,x), range(-4 4)
Same as above, but add the normal probability density function
     twoway function y = tden(4,x), range(-4 4) ||
                                                          ///
          function y = normalden(x), range(-4 4)
Add a legend
     twoway function y = tden(4,x), range(-4 4) ||
                                                         ///
           function y = normalden(x), range(-4 4)
                                                         ///
          legend(label(1 "t density with 4 df")
                                                         ///
          label(2 "Normal density"))
Normal probability density function with lines from the curve to 0 at -1.96 and 1.96
     twoway function y = normalden(x), range(-4 4) dropline(-1.96 1.96)
```

Menu

```
Graphics > Twoway graph (scatter, line, etc.)
```

2

Syntax

 $\underline{\text{tw}}$ oway function [[y] =] f(x) [if] [in] [, options]

options	Description
range(# #) range(varname) n(#)	plot over x = # to # plot over x = min to max of <i>varname</i> evaluate at # points; default is 300
<pre>droplines(numlist) base(#)</pre>	draw lines to axis at specified x values base value for dropline(); default is 0
<u>hor</u> izontal	draw plot horizontally
<pre>yvarformat(%fmt) xvarformat(%fmt)</pre>	display format for y display format for x
cline_options	change look of plotted line
axis_choice_options	associate plot with alternative axis
twoway_options	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

All explicit options are *rightmost*, except horizontal, which is *unique*; see [G-4] **Concept: repeated options**.

if exp and in range play no role unless option range (varname) is specified.

In the above syntax diagram, f(x) stands for an expression in terms of x.

Options

range(##) and range(varname) specify the range of values for x. In the first syntax, range() is
a pair of numbers identifying the minimum and maximum. In the second syntax, range() is a
variable name, and the range used will be obtained from the minimum and maximum values of
the variable. If range() is not specified, range(0 1) is assumed.

n(#) specifies the number of points at which f(x) is to be evaluated. The default is n(300).

droplines (numlist) adds dropped lines from the function down to, or up to, the axis (or y = base() if base() is specified) at each x value specified in numlist.

base(#) specifies the base for the droplines(). The default is base(0). This option does not affect the range of the axes, so you may also want to specify the axis_scale_option yscale(range(#)) as well; see [G-3] axis_scale_options.

horizontal specifies that the roles of y and x be interchanged and that the graph be plotted horizontally rather than vertically (that the plotted function be reflected along the identity line).

yvarformat(%fmt) and xvarformat(%fmt) specify the display format to be used for y and x. These formats are used when labeling the axes; see [G-3] axis_label_options.

cline_options specify how the function line is rendered; see [G-3] cline_options.

 $axis_choice_options$ associate the plot with a particular y or x axis on the graph; see [G-3] $axis_choice_options$.

twoway_options are a set of common options supported by all twoway graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over by () groups, and change some advanced settings. See [G-3] twoway_options.

Remarks and examples

stata.com


Remarks are presented under the following headings:

Typical use Advanced use 1 Advanced use 2

Typical use


You wish to plot the function $y = \exp(-x/6)\sin(x)$ over the range 0 to 4π :

. twoway function $y=\exp(-x/6)*\sin(x)$, range(0 12.57)



A better rendition of the graph above is

. twoway function y=exp(-x/6)*sin(x), range(0 12.57)
 yline(0, lstyle(foreground))
 xlabel(0 3.14 "{&pi}" 6.28 "2{&pi}" 9.42 "3{&pi}" 12.57 "4{&pi}")
 xsca(noline)



yline(0, lstyle(foreground)) added a line at y = 0; lstyle(foreground) gave the line the same style as used for the axes. See [G-3] *added_line_options*.

xlabel(0 3.14 "{&pi}" 6.28 "2{&pi}" 9.42 "3{&pi}" 12.57 "4{&pi}") labeled the x axis with the numeric values given; see [G-3] $axis_label_options$.

xsca(noline) suppressed the drawing of the x-axis line; see [G-3] axis_scale_options.

3 SD

4 SD

Advanced use 1

The following graph appears in many introductory textbooks:

-2 SD

-1 SD

-4 SD

-3 SD

```
. twowav
     function y=normalden(x), range(-4 -1.96) color(gs12) recast(area)
 function y=normalden(x), range(1.96 4) color(gs12) recast(area)
 function y=normalden(x), range(-4 4) lstyle(foreground)
 ΙΙ,
     ysca(off) xsca(noline)
     legend(off)
     xlabel(-4 "-4 SD" -3 "-3 SD" -2 "-2 SD" -1 "-1 SD" 0 "Mean"
             1 "1 SD" 2 "2 SD" 3 "3 SD" 4 "4 SD"
     , gmin gmax)
     xtitle("")
```

We drew the graph in three parts: the shaded area on the left, the shaded area on the right, and then the overall function. To obtain the shaded areas, we used the advanced_option recast(area) so that, rather than the function being plotted by graph twoway line, it was plotted by graph twoway area; see [G-3] advanced_options and [G-2] graph twoway area. Concerning the overall function, we drew it last so that its darker foreground-colored line would not get covered up by the shaded areas.

Mean


1 SD

2 SD

Advanced use 2

function plots may be overlaid with other twoway plots. For instance, function is one way to add y=x lines to a plot:

- . use https://www.stata-press.com/data/r18/sp500, clear
 (S&P 500)
- . scatter open close, msize(*.25) mcolor(*.6) ||
 function y=x, range(close) yvarlab("y=x")



In the above, we specified the *advanced_option* yvarlab("y=x") so that the variable label of y would be treated as "y=x" in the construction of the legend; see [G-3] *advanced_options*. We specified msize(*.25) to make the marker symbols smaller, and we specified mcolor(*.6) to make them dimmer; see [G-4] *size* and [G-4] *colorstyle*.

Reference

Cox, N. J. 2004. Stata tip 15: Function graphs on the fly. Stata Journal 4: 488-489.

Also see

[G-2] graph twoway line — Twoway line plots⁺

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on citing Stata documentation.